



Univerza v Ljubljani

Fakulteta za računalništvo in informatiko

Tržaška 25, Ljubljana

**Poročilo iz vaj pri predmetu
MODELIRANJE IN SIMULACIJE**

Ljubljana, 22. 5. 2000

Mitja Luštrek
Kemal Posedi

NALOGA

Napiši program za simulacijo dvigal v poslovni stavbi. V njej je 5 nadstropij in 4 dvigala.

Čas za premik iz mirovanja za 1 nadstropje je 10 ± 1 s. Enak je čas zaviranja. Brez ustavljanja prevozi dvigalo nadstropje v 6 ± 1 s.

V pritličju vstopi 400 ± 150 oseb na uro. Enakomerno se porazdelijo po nadstropjih. 50 jih ponovi vožnjo med nadstropji enkrat, 20 dvakrat in 7 trikrat.

Vse porazdelitve so normalne.

Simuliraj delovanje dvigal in ugotovi potrebno kapaciteto, da čakanje ne bo daljše od 100 s.

REŠITEV

Osebe prihajajo v pritličje vsake $9 \pm 3,5$ s (kar približno ustreza danim podatkom, saj točna pretvorba iz oseb na časovno enoto v časovno enoto na osebo ni možna). Tu so vštete osebe, ki ponovijo vožnjo dvakrat ali trikrat (skupaj 27 na uro) – obravnavava jih namreč, kot da bi prišle na novo, saj z vidika dvigal tudi je tako. V pritličje se vozijo vsakih 43 s (tu so enkrat štete osebe, ki vožnjo ponovijo enkrat ali dvakrat, in dvakrat tiste, ki jo ponovijo trikrat – skupaj 84 oseb na uro). Obravnavava jih kot na novo nastale osebe, ker z vidika dvigala ni važno, da so se enkrat prej peljale gor. Izkaže se, da je potrebna kapaciteta dvigal 4 osebe. Sicer je čakalni čas bistveno krajši od zahtevanega, vendar je narava sistema tako, da če dvigala oseb ne morejo zvoziti sproti, v daljšem obdobju čakalni čas naraste čez vse meje – pri kapaciteti 3 osebe se v 8 urah že nabere preveč čakajočih.

Program generira osebe, ki skušajo zaseči *storage* kapacitete enega dvigala. Kadar v pritličju ni dvigala, je ta *storage* izključen, kadar pa dvigalo pride, se vključi (osebe, ki vanj vstopijo, ga takoj spet izključijo). To skrbi, da vedno vstopijo po 4 osebe naenkrat in da lahko opazujemo vrsto pred dvigali. Ko osebe vstopijo, nadstropja, kamor želijo iti, zapišejo v spremenljivke, nato pa se terminirajo (ker niso več potrebne). 4 dvigala se zgenerirajo v svojem bloku. Vsako, ko pride v pritličje, te spremenljivke prebere v svoje parametre, nato pa jih nastavi na začetno vrednost (da so nared za vpis novih vrednosti, ko pride naslednje dvigalo). Nato odpotuje do najvišjega nadstropja, za katerega ima potnike oziroma kjer ga potniki čakajo. Na poti navzgor odlaga potnike, na poti navzdol pa jih sprejema. Navzdol namenjeni potniki se generirajo v svojem bloku, nato pa se jim priredi nadstropje, iz katerega gredo.

KODA

```

10 SNORM FUNCTION RN4,C25 ;Normalna porazdelitev
    na [-5, 5]
0,-5/.00003,-4/.00135,-3/.00621,-2.5/.02275,-2
.06681,-1.5/.11507,-1.2/.15866,-1/.21186,-.8/.27425,-.6
.34458,-.4/.42074,-.2/.5,0/.57926,.2/.65542,.4
.72575,.6/.78814,.8/.84134,1/.88493,1.2/.93319,1.5
.97725,2/.99379,2.5/.99865,3/.99997,4/1,5
20 Ljudje FVARIABLE 0.7#FN$SNORM ;Funk. za generiranje
    ljudi
21 EnaSek FVARIABLE 0.2#FN$SNORM ;Funk. za +/-1 s
30 Nadstropje FUNCTION RN1,D5 ;Funk. za dolocanje
    nadstropja
0.2,1/0.4,2/0.6,3/0.8,4/1,5
40 INITIAL X$C11,-1 ;Nadstropje 1. cloveka
41 INITIAL X$C12,-1 ;Nadstropje 2. cloveka
42 INITIAL X$C13,-1 ;Nadstropje 3. cloveka
43 INITIAL X$C14,-1 ;Nadstropje 4. cloveka
50 Dostop STORAGE 4 ;Dostop do dvigal

100 GENERATE 9,X$Ljudje ;Dela ljudi, ki gredo
    gor.
110 Cakaj QUEUE Vrsta ;Ljudje vstopajo v
    vrsto.
111 ENTER Dostop ;Ljudje vstopijo v
    dvigalo.
112 DEPART Vrsta ;Ljudje zapustijo vrsto.
113 ADVANCE 1 ;Pocaka, da vstopijo
    trije.
114 SUNAVAIL Dostop ;Zapre dostop do
    dvigala.
115 LEAVE Dostop ;Ljudje zapustijo vhod v
    dvigalo.
120 TEST E X$C11,-1,Polni2 ;Ima 1. clovek
    nadstropje? Ne: naprej, da: na 2. cloveka.
121 SAVEVALUE C11, FN$Nadstropje ;1. clovek dobi
    nadstropje.
122 TERMINATE ;1. clovek umre.
130 Polni2 TEST E X$C12,-1,Polni3 ;Ima 2. clovek
    nadstropje? Ne: naprej, da: na 3. cloveka.
131 SAVEVALUE C12, FN$Nadstropje ;2. clovek dobi
    nadstropje.
132 TERMINATE ;2. clovek umre.
140 Polni3 TEST E X$C13,-1,Polni4 ;Ima 3. clovek
    nadstropje? Ne: naprej, da: na zacetek (ce pride do napake).
141 SAVEVALUE C13, FN$Nadstropje ;3. clovek dobi
    nadstropje.
142 TERMINATE ;3. clovek umre.
150 Polni4 TEST E X$C14,-1,Cakaj ;Ima 4. clovek
    nadstropje? Ne: naprej, da: na zacetek (ce pride do napake).
151 SAVEVALUE C14, FN$Nadstropje ;4. clovek dobi
    nadstropje.
152 TERMINATE ;4. clovek umre.

;P1 ... 1. clovek
;P2 ... 2. clovek
;P3 ... 3. clovek
;P4 ... 4. clovek
;P10 ... trenutno nadstropje

200 GENERATE ,,,4 ;Naredi dvigala
210 Odpri SAVAIL Dostop ;Odpre dostop do
    dvigala.
211 ADVANCE 2 ;Pocaka, da ljudje
    vstopijo.
220 TEST G X$C11,-1,Odpri ;Je 1. clovek (vedno
    najprej vstopi 1.) notri? Da: naprej, ne: cakaj v zanki.
230 ASSIGN 1,X$C11 ;Nadstropje 1. cloveka
    zapise v parameter dvigala.
231 ASSIGN 2,X$C12 ;Nadstropje 2. cloveka
    zapise v parameter dvigala.
232 ASSIGN 3,X$C13 ;Nadstropje 3. cloveka
    zapise v parameter dvigala.

```

```

233 ASSIGN 4,X$C14 ;Nadstropje 4. cloveka
zapise v parameter dvigala.
240 SAVEVALUE C11,-1 ;Zbrise nadstropje 1.
cloveka.
241 SAVEVALUE C12,-1 ;Zbrise nadstropje 2.
cloveka.
242 SAVEVALUE C13,-1 ;Zbrise nadstropje 3.
cloveka.
243 SAVEVALUE C14,-1 ;Zbrise nadstropje 4.
cloveka.
250 ADVANCE 10,X$EnaSek ;Dvigalo spelje iz
pritlicja.

300 Gor ASSIGN 10+,1 ;Dvigalo se premakne za
1 nadstropje gor.
301 TEST NE P1,P10,Stoj ;Je 1. clovek namenjen v
trenutno nadstropje? Ne: naprej, da: ustavi.
302 TEST NE P2,P10,Stoj ;Je 2. clovek namenjen v
trenutno nadstropje? Ne: naprej, da: ustavi.
303 TEST NE P3,P10,Stoj ;Je 3. clovek namenjen v
trenutno nadstropje? Ne: naprej, da: ustavi.
304 TEST NE P4,P10,Stoj ;Je 4. clovek namenjen v
trenutno nadstropje? Ne: naprej, da: ustavi.
310 ADVANCE 6,X$EnaSek ;Dvigalo prevozi
nadstropje.
311 TRANSFER ,Testi ;Gre preverjat, ce mora
se peljati gor.
320 Stoj ADVANCE 10,X$EnaSek ;Dvigalo se ustavi.
321 ADVANCE 10,X$EnaSek ;Dvigalo spelje.
330 Testi TEST LE P1,P10,Gor ;Je 1. clovek namenjen v
nadstropje nad trenutnim? Ne: naprej, da: pelji gor.
331 TEST LE P2,P10,Gor ;Je 2. clovek namenjen v
nadstropje nad trenutnim? Ne: naprej, da: pelji gor.
332 TEST LE P3,P10,Gor ;Je 3. clovek namenjen v
nadstropje nad trenutnim? Ne: naprej, da: pelji gor.
333 TEST LE P4,P10,Gor ;Je 4. clovek namenjen v
nadstropje nad trenutnim? Ne: naprej, da: pelji gor.
340 TEST NE P10,1,Sem1 ;Je dvigalo v 1.
nadstropju? Ne: naprej, da: pojdi preverjat stanje nad 1. nadstropjem.
341 TEST NE P10,2,Sem2 ;Je dvigalo v 2.
nadstropju? Ne: naprej, da: pojdi preverjat stanje nad 2. nadstropjem.
342 TEST NE P10,3,Sem3 ;Je dvigalo v 3.
nadstropju? Ne: naprej, da: pojdi preverjat stanje nad 3. nadstropjem.
343 TEST NE P10,4,Sem4 ;Je dvigalo v 4.
nadstropju? Ne: naprej, da: pojdi preverjat stanje nad 4. nadstropjem.
344 TRANSFER ,Dol ;Ocitno je dvigalo v 5.
nadstropju, zato pelji dol.
350 Sem1 TEST L X$Caka2,1,Gor ;V 2. nadstropju kdo
caka? Ne: naprej, da: pelji gor.
351 Sem2 TEST L X$Caka3,1,Gor ;V 3. nadstropju kdo
caka? Ne: naprej, da: pelji gor.
352 Sem3 TEST L X$Caka4,1,Gor ;V 4. nadstropju kdo
caka? Ne: naprej, da: pelji gor.
353 Sem4 TEST L X$Caka5,1,Gor ;V 5. nadstropju kdo
caka? Ne: naprej, da: pelji gor.
360 TRANSFER ,Dol ;Pelji dol.

400 Dol TEST NE P10,5,Sem5dol ;Je dvigalo v 5.
nadstropju? Ne: naprej, da: pojdi pogledat, ce tam kdo caka.
401 TEST NE P10,4,Sem4dol ;Je dvigalo v 4.
nadstropju? Ne: naprej, da: pojdi pogledat, ce tam kdo caka.
402 TEST NE P10,3,Sem3dol ;Je dvigalo v 3.
nadstropju? Ne: naprej, da: pojdi pogledat, ce tam kdo caka.
403 TEST NE P10,2,Sem2dol ;Je dvigalo v 2.
nadstropju? Ne: naprej, da: pojdi pogledat, ce tam kdo caka.
404 TRANSFER ,Sem1dol ;Ocitno je dvigalo v 1.
nadstropju, zato pogledat, ce tam kdo caka.
410 Sem5dol TEST G X$Caka5,0,Naprej ;V 5. nadstropju kdo
caka? Da: naprej, ne: pelji naprej.
411 SAVEVALUE Caka5,0 ;V 5. nadstropju ne caka
nihce vec.
412 TRANSFER ,StojDol ;Pojdi na ustavljanje
dvigala.
420 Sem4dol TEST G X$Caka4,0,Naprej ;V 4. nadstropju kdo
caka? Da: naprej, ne: pelji naprej.

```

```

421 SAVEVALUE Caka4,0 ;V 4. nadstropju ne caka
nihce vec.
422 TRANSFER ,StojDol ;Pojdi na ustavljanje
dvigala.
430 Sem3dol TEST G X$Caka3,0,Naprej ;V 3. nadstropju kdo
caka? Da: naprej, ne: pelji naprej.
431 SAVEVALUE Caka3,0 ;V 3. nadstropju ne caka
nihce vec.
432 TRANSFER ,StojDol ;Pojdi na ustavljanje
dvigala.
440 Sem2dol TEST G X$Caka2,0,Naprej ;V 2. nadstropju kdo
caka? Da: naprej, ne: pelji naprej.
441 SAVEVALUE Caka2,0 ;V 2. nadstropju ne caka
nihce vec.
442 TRANSFER ,StojDol ;Pojdi na ustavljanje
dvigala.
450 Sem1dol TEST G X$Caka1,0,Naprej ;V 1. nadstropju kdo
caka? Da: naprej, ne: pelji naprej.
451 SAVEVALUE Caka1,0 ;V 1. nadstropju ne caka
nihce vec.
460 StojDol ADVANCE 10,X$EnaSek ;Dvigalo se ustavi.
461 ADVANCE 10,X$EnaSek ;Dvigalo spelje.
462 TRANSFER ,JeKonec ;Gre preverjat, ce je
dvigalo ze v 1. nadstropju.
470 Naprej ADVANCE 6,X$EnaSek ;Dvigalo prevozi
nadstropje.
480 JeKonec ASSIGN 10-,1 ;Dvigalo se premakne za
1 nadstropje dol.
481 TEST E P10,0,Dol ;Je ze v 0. nadstropju?
Da: naprej, ne: gre na zacetek voznje dol.
490 TRANSFER ,Odpri ;Gre spustit nove ljudi
v dvigalo.

500 GENERATE 43 ;Dela ljudi, ki gredo
dol.
510 SAVEVALUE OdKod,FN$Nadstropje ;Clovek dobi nadstropje.
520 TEST NE X$OdKod,2,Iz2 ;Je to 2. nadstropje?
Ne: naprej, da: pojdi na dolocanje stevila cakajocih v 2. nadstropju.
521 TEST NE X$OdKod,3,Iz3 ;Je to 3. nadstropje?
Ne: naprej, da: pojdi na dolocanje stevila cakajocih v 3. nadstropju.
522 TEST NE X$OdKod,4,Iz4 ;Je to 4. nadstropje?
Ne: naprej, da: pojdi na dolocanje stevila cakajocih v 4. nadstropju.
523 TEST NE X$OdKod,5,Iz5 ;Je to 5. nadstropje?
Ne: naprej, da: pojdi na dolocanje stevila cakajocih v 5. nadstropju.
530 SAVEVALUE Caka2+,1 ;Ocitno je to 1.
nadstropje, zato v 1. nadstropju caka 1 clovek vec.
531 TERMINATE ;Clovek umre.
540 Iz2 SAVEVALUE Caka2+,1 ;V 2. nadstropju caka 1
clovek vec.
541 TERMINATE ;Clovek umre.
550 Iz3 SAVEVALUE Caka3+,1 ;V 3. nadstropju caka 1
clovek vec.
551 TERMINATE ;Clovek umre.
560 Iz4 SAVEVALUE Caka4+,1 ;V 4. nadstropju caka 1
clovek vec.
561 TERMINATE ;Clovek umre.
570 Iz5 SAVEVALUE Caka5+,1 ;V 5. nadstropju caka 1
clovek vec.
571 TERMINATE ;Clovek umre.

1000 GENERATE 28800 ;Koliko casa na traja
simulacija.
1010 TERMINATE 1 ;Simulacija se konca.

```

POROČILO

GPSS/PC Report file REPORT.GPS. (V 2, # 40390.21875) 05-21-2000 23: page 1

START_TIME	END_TIME	BLOCKS	FACILITIES	STORAGES	FREE_MEMORY
0	28800	99	0	1	896

LINE	LOC	BLOCK_TYPE	ENTRY_COUNT	CURRENT_COUNT	RETRY
100	1	GENERATE	3200	0	0
110	CAKAJ	QUEUE	3206	2	0
111	3	ENTER	3204	0	0
112	4	DEPART	3204	0	0
113	5	ADVANCE	3204	0	0
114	6	SUNAVAIL	3204	0	0
115	7	LEAVE	3204	0	0
120	8	TEST	3204	0	0
121	9	SAVEVALUE	1119	0	0
122	10	TERMINATE	1119	0	0
130	POLNI2	TEST	2085	0	0
131	12	SAVEVALUE	839	0	0
132	13	TERMINATE	839	0	0
140	POLNI3	TEST	1246	0	0
141	15	SAVEVALUE	700	0	0
142	16	TERMINATE	700	0	0
150	POLNI4	TEST	546	0	0
151	18	SAVEVALUE	540	0	0
152	19	TERMINATE	540	0	0
200	20	GENERATE	4	0	0
210	ODPRI	SAVAIL	1407	0	0
211	22	ADVANCE	1407	0	0
220	23	TEST	1407	0	0
230	24	ASSIGN	1119	0	0
231	25	ASSIGN	1119	0	0
232	26	ASSIGN	1119	0	0
233	27	ASSIGN	1119	0	0
240	28	SAVEVALUE	1119	0	0
241	29	SAVEVALUE	1119	0	0
242	30	SAVEVALUE	1119	0	0
243	31	SAVEVALUE	1119	0	0
250	32	ADVANCE	1119	0	0
300	GOR	ASSIGN	4774	0	0
301	34	TEST	4774	0	0
302	35	TEST	3655	0	0
303	36	TEST	2964	0	0
304	37	TEST	2526	0	0
310	38	ADVANCE	2258	0	0
311	39	TRANSFER	2258	0	0
320	STOJ	ADVANCE	2516	1	0
321	41	ADVANCE	2515	2	0
330	TESTI	TEST	4771	0	0
331	43	TEST	2490	0	0
332	44	TEST	1791	0	0
333	45	TEST	1546	0	0
340	46	TEST	1432	0	0

GPSS/PC Report file REPORT.GPS. (V 2, # 40390.21875) 05-21-2000 23: page 2

LINE	LOC	BLOCK_TYPE	ENTRY_COUNT	CURRENT_COUNT	RETRY
341	47	TEST	1379	0	0
342	48	TEST	1239	0	0
343	49	TEST	1004	0	0
344	50	TRANSFER	596	0	0
350	SEM1	TEST	53	0	0
351	SEM2	TEST	188	0	0
352	SEM3	TEST	388	0	0
353	SEM4	TEST	695	0	0
360	55	TRANSFER	520	0	0
400	DOL	TEST	4765	0	0
401	57	TEST	4169	0	0
402	58	TEST	3253	0	0
403	59	TEST	2211	0	0
404	60	TRANSFER	1115	0	0
410	SEM5DOL	TEST	596	0	0
411	62	SAVEVALUE	130	0	0
412	63	TRANSFER	130	0	0
420	SEM4DOL	TEST	916	0	0
421	65	SAVEVALUE	120	0	0
422	66	TRANSFER	120	0	0
430	SEM3DOL	TEST	1042	0	0
431	68	SAVEVALUE	119	0	0
432	69	TRANSFER	119	0	0
440	SEM2DOL	TEST	1096	0	0
441	71	SAVEVALUE	125	0	0
442	72	TRANSFER	125	0	0
450	SEM1DOL	TEST	1115	0	0
451	74	SAVEVALUE	132	0	0
460	STOJDOL	ADVANCE	626	0	0
461	76	ADVANCE	626	1	0
462	77	TRANSFER	625	0	0
470	NAPREJ	ADVANCE	4139	0	0
480	JEKONEC	ASSIGN	4764	0	0
481	80	TEST	4764	0	0
490	81	TRANSFER	1115	0	0
500	82	GENERATE	669	0	0
510	83	SAVEVALUE	669	0	0
520	84	TEST	669	0	0
521	85	TEST	535	0	0
522	86	TEST	409	0	0
523	87	TEST	284	0	0
530	88	SAVEVALUE	139	0	0
531	89	TERMINATE	139	0	0
540	IZ2	SAVEVALUE	134	0	0
541	91	TERMINATE	134	0	0
550	IZ3	SAVEVALUE	126	0	0
551	93	TERMINATE	126	0	0
560	IZ4	SAVEVALUE	125	0	0
561	95	TERMINATE	125	0	0
570	IZ5	SAVEVALUE	145	0	0

GPSS/PC Report file REPORT.GPS. (V 2, # 40390.21875) 05-21-2000 23: page 3

LINE	LOC	BLOCK_TYPE	ENTRY_COUNT	CURRENT_COUNT	RETRY
571	97	TERMINATE	145	0	0
1000	98	GENERATE	1	0	0
1010	99	TERMINATE	1	0	0

QUEUE	MAX	CONT.	ENTRIES	ENTRIES(0)	AVE.CONT.	AVE.TIME	AVE.(-0)	RETRY
VRSTA	11	2	3206	199	3.23	28.99	30.90	0

STORAGE	CAP.	REMAIN.	MIN.	MAX.	ENTRIES	AVL.	AVE.C.	UTIL.	RETRY	DELAY
DOSTOP	4	4	0	4	3204	0	0.11	0.028	0	2

SAVEVALUE	VALUE	RETRY
LJUDJE	+0	0
ENASEK	+0	0
CL1	-1	0
CL2	-1	0
CL3	-1	0
CL4	-1	0
CAKA2	+0	0
CAKA3	+0	0
CAKA4	+0	0
CAKA5	+0	0
CAKA1	+0	0
ODKOD	+3	0